
     H dftactgrp(*NO) bnddir('WEBSERVICE')

     Fweathrspfmcf   e             workstn

     F                                     sfile(sfl:rrn)

     F                                     indds(dspindicators)

      /copy /WebServices/weatherws/WeatherSoap.rpgleinc

     d WebServiceDs    ds                  likeds(This_t)

     d ZipCodeDs       ds                  likeds(xsd_string)

     d ForeCastDs      ds                  likeds(ForecastReturn_t)

     D rrn             s              4  0

     D workyear        s              2  0

     D workdays        s              3  0

     D workdatejul     s              5  0

     D displaydate     s               d   datfmt(*iso)

     D

     D dspindicators   DS

     D  hourly...

     D  requested              1      1n

     D  endofjob               3      3n

     D  return                12     12n

     D  subfileclear          60     60n

     D  subfiledsp            62     62n

     D  zipcodezero           71     71n

     D  zipcodenot...

     D found                  72     72n

     D webservicedown         73     73n

     D zipcodealpha    S              5

     D*--------------------------------------------------

     D* Program name: DisplayHourly

     D* Purpose:        Launches www.weather.com with selected zip code

     D* Returns:

     D*--------------------------------------------------

     D DisplayHourly   PR                  EXTPGM('HOURLYCL')

     D  zipcodealpha                  5    const

      /free

       //  Prompt user for zip code

          dou endofjob;

          exfmt screen1;

       //  Was EOJ requested?

           if endofjob;

            *inlr = *on;

            return;

           endif;

       //  Clear error conditons

           zipcodezero = *off;

           zipcodenotfound = *off;

           webservicedown = *off;

       //  Zipcode cannot be equal to all zeros

          if zipcode = *zeros;

           zipcodezero = *on;

           iter;

           endif;

       //  Prepare to call soap service

         clear WebServiceDs;

         clear ZipCodeDs;

         clear ForecastDs;

       //ZipCodeDs.value = %char(zipcode);

         ZipCodeDs.value = %editc(zipcode:'X');

       //  Allocate space in storage

            if stub_create_WeatherSoap(WebServiceDs);

        // Invoke the Web service operation.

       if stub_op_GetCityForecastByZip(WebServiceDs:ZipCodeDs:ForecastDs);

          endif;

        // Destroy Web service stubs.

         stub_destroy_WeatherSoap(WebServiceDs);

           endif;

        // Check if an error occurred during call to service

          if webserviceds.excoccurred or webserviceds.excstring > *blanks;

           webservicedown = *on;

            iter;

          endif;

       //  When here, we got some data!

       //  See if we have success (value of '1' will be returned)

           if forecastds.success.value = 0;

           zipcodenotfound = *on;

          iter;

        endif;

       //  If we made it this far, good data has been received

          city = forecastds.city.value;

          state = forecastds.state.value;

          wthrsta = forecastds.weatherstationcity.value;

       //  Prepare subfile control record

         rrn = *zeros;

         subfileclear = *on;

         subfiledsp = *off;

         write sflctl;

         subfileclear = *off;

         subfiledsp = *on;

       //  Capture detail data, load subfile

        for rrn = 1 to 7;

       //   wthrdate = %subst(weatherdata.forecast(rrn).date:1:10);

       // Description

        desc = forecastds.forecastresult.forecast...

        _ref.array(rrn).desciption.value;

       // Morning Low

        templo = forecastds.forecastresult.forecast...

        _ref.array(rrn).temperatures.morninglow.value;

       // Daytime High

        temphi = forecastds.forecastresult.forecast...

        _ref.array(rrn).temperatures.daytimehigh.value;

       // Precipitation Morning

        precpmo = forecastds.forecastresult.forecast...

        _ref.array(rrn).probabilityofprecipiation.daytime.value;

       // Precipitation Nighttime

        precpnt = forecastds.forecastresult.forecast...

        _ref.array(rrn).probabilityofprecipiation.nighttime.value;

       // Returned days will be days from January 1st (0-365)

        workdays  = forecastds.forecastresult.forecast...

        _ref.array(rrn).date.value.yday;

       // Returned year will be years from 1900, need to subtract 100

        workyear  = forecastds.forecastresult.forecast...

        _ref.array(rrn).date.value.year - 100;

       // Calculate julian work date.  If valid move to display date, else move *loval.

        workdatejul = (workyear * 1000) + workdays;

           test(de)  *jul workdatejul;

           if not %error;

             displaydate = %date(workdatejul:*jul);

             wthrdate = %char(displaydate);

           else;

             wthrdate = '01-01-0001';

           endif;

        write sfl;

        endfor;

       //  Display subfile.  If F1 is selected display weather.com

          dou return;

           exfmt sflctl;

             if hourlyrequested;

               zipcodealpha = %editc(zipcode:'X');

               displayhourly(zipcodealpha);

             endif;

          enddo;

            enddo;


